The partition factorial constant
and asymptotics of the sequence A058694

(*Václav Kotěšovec, published Jun 26 2015*)

The sequence A058694 is a partial product $p(1) * \dots * p(n)$ of partition numbers A000041.
Main result:

$$\prod_{k=1}^{n} p(k) \sim C * \frac{\Gamma\left(\frac{23}{24}\right)}{n^{n+\frac{11}{24}+\frac{3}{4\pi^2}}\ 2^{2n}\ 3^{n/2}\ \sqrt{2\pi}} * exp\left(\pi\left(\frac{2n}{3}\right)^{3/2} + n + \left(\frac{11\,\pi}{12\sqrt{6}} - \frac{\sqrt{6}}{\pi}\right)\sqrt{n} + S\right)$$

where

$$C = \lim_{n\to\infty} \prod_{k=1}^{n} \frac{p(k)}{\frac{\exp\left(\pi\sqrt{2\,(k-1/24)/3}\right)}{4\sqrt{3}\,(k-1/24)} * \left(1 - \frac{1}{\pi}\sqrt{\frac{3}{2\,(k-1/24)}}\right)}$$

A proposed name for C is "**partition factorial constant**"

C = A259314 = 0.91101673133224995186154746959468345278073860978008093028132149022759149124 ...

$$S = \pi\sqrt{\frac{2}{3}}\,\zeta\left(-\frac{1}{2},\frac{23}{24}\right) - \frac{1}{\pi}\sqrt{\frac{3}{2}}\,\zeta\left(\frac{1}{2},\frac{23}{24}\right) + \frac{3\Gamma'\left(\frac{23}{24}\right)}{4\pi^2\,\Gamma\left(\frac{23}{24}\right)} - \sum_{j=3}^{\infty}\zeta\left(\frac{j}{2},\frac{23}{24}\right) * \frac{1}{j} * \left(\frac{1}{\pi}\sqrt{\frac{3}{2}}\right)^{j}$$

S = −0.025419333977936527099030120192256408130475739685794740695249403875762913263497141 ...

ζ is the Hurwitz Zeta Function (in Maple notation Zeta(0,z,v), in Mathematica Zeta[z,v], equivalently HurwitzZeta[z,v]).

Proof:

The asymptotics of partition function by Hardy and Ramanujan is well known [see 2], [3],

$$p(n) \sim \frac{\exp\left(\pi\sqrt{\frac{2n}{3}}\right)}{4n\sqrt{3}}$$

but we need more precise asymptotics [see 4, page 90]

$$p(n) \sim \frac{\exp\left(\pi\sqrt{\frac{2}{3}\left(n-\frac{1}{24}\right)}\right)}{4\sqrt{3}\left(n-\frac{1}{24}\right)} * \left(1 - \frac{1}{\pi}\sqrt{\frac{3}{2\left(n-\frac{1}{24}\right)}}\right)$$

We will now analyze three products

Product 1

$$\prod_{k=1}^{n} \exp\left(\pi \sqrt{\frac{2}{3}\left(k-\frac{1}{24}\right)}\right)$$

```
Product[E^(Sqrt[2/3] Sqrt[k - 1/24] π), {k, 1, n}]
```

$$e^{\sqrt{\frac{2}{3}}\,\pi\left(\text{HurwitzZeta}\left[-\frac{1}{2},\frac{23}{24}\right]-\text{HurwitzZeta}\left[-\frac{1}{2},\frac{23}{24}+n\right]\right)}$$

```
Series[Sqrt[2/3] π (HurwitzZeta[-1/2, 23/24] - HurwitzZeta[-1/2, 23/24 + n]), {n, Infinity, 2}]
```

$$\frac{2}{3}\sqrt{\frac{2}{3}}\,\pi\, n^{3/2}+\frac{11\,\pi\sqrt{n}}{12\sqrt{6}}+\sqrt{\frac{2}{3}}\,\pi\,\text{HurwitzZeta}\left[-\frac{1}{2},\frac{23}{24}\right]+\frac{73\,\pi\sqrt{\frac{1}{n}}}{1152\sqrt{6}}+\frac{253\,\pi\left(\frac{1}{n}\right)^{3/2}}{165\,888\sqrt{6}}+O\left[\frac{1}{n}\right]^{5/2}$$

$$\prod_{k=1}^{n} \exp\left(\pi \sqrt{\frac{2}{3}\left(k-\frac{1}{24}\right)}\right) \sim \exp\left(\frac{2}{3}\sqrt{\frac{2}{3}}\pi\, n^{3/2}+\frac{11\pi\sqrt{n}}{12\sqrt{6}}+\sqrt{\frac{2}{3}}\pi\,\zeta\left(-\frac{1}{2},\frac{23}{24}\right)\right)$$

Product 2

$$\prod_{k=1}^{n} \frac{1}{4\sqrt{3}\left(k-\frac{1}{24}\right)}$$

```
Product[1/(4 Sqrt[3] (k - 1/24)), {k, 1, n}]
```

$$\frac{3^{-n/2}\,4^{-n}\,\text{Gamma}\left[\frac{23}{24}\right]}{\text{Gamma}\left[\frac{23}{24}+n\right]}$$

```
FullSimplify[Normal[Series[(3^(-n/2) 4^-n Gamma[23/24])/Gamma[23/24 + n], {n, Infinity, 2}]], n > 0]
```

$$\frac{2^{-\frac{31}{2}-2n}\,3^{-4-\frac{n}{2}}\,e^{n}\,n^{-\frac{59}{24}-n}\,(-2767+2304\,n\,(-73+1152\,n))\,\text{Gamma}\left[\frac{23}{24}\right]}{\sqrt{\pi}}$$

```
FullSimplify[(2^(-31/2 - 2 n) 3^(-4 - n/2) E^n n^(-59/24 - n) (2304*1152*n^2) Gamma[23/24])/Sqrt[π]]
```

$$\frac{2^{-\frac{1}{2}-2n}\,3^{-n/2}\,e^{n}\,n^{-\frac{11}{24}-n}\,\text{Gamma}\left[\frac{23}{24}\right]}{\sqrt{\pi}}$$

$$\prod_{k=1}^{n} \frac{1}{4\sqrt{3}\left(k-\frac{1}{24}\right)} \sim \frac{2^{-2n-\frac{1}{2}}\,3^{-\frac{n}{2}}\,e^{n}\,n^{-n-\frac{11}{24}}\,\Gamma\left(\frac{23}{24}\right)}{\sqrt{\pi}}$$

Product 3

$$\prod_{k=1}^{n}\left(1-\frac{1}{\pi}\sqrt{\frac{3}{2\left(k-\frac{1}{24}\right)}}\right)$$

$$\log\left(\prod_{k=1}^{n}\left(1-\frac{1}{\pi}\sqrt{\frac{3}{2\left(k-\frac{1}{24}\right)}}\right)\right)=\sum_{k=1}^{n}\log\left(1-\frac{1}{\pi}\sqrt{\frac{3}{2\left(k-\frac{1}{24}\right)}}\right)=-\sum_{k=1}^{n}\sum_{j=1}^{\infty}\frac{1}{j}\left(\frac{1}{\pi}\sqrt{\frac{3}{2\left(k-\frac{1}{24}\right)}}\right)^{j}=-\sum_{j=1}^{\infty}\frac{3^{j/2}}{j\,\pi^{j}\,2^{j/2}}\sum_{k=1}^{n}\frac{1}{\left(k-\frac{1}{24}\right)^{j/2}}$$

```
Sum[1 / (k - 1 / 24) ^ (j / 2), {k, 1, n}]
HurwitzZeta[j/2, 23/24] - HurwitzZeta[j/2, 23/24 + n]
```

Case $j=1$

```
Series[HurwitzZeta[1/2, 23/24] - HurwitzZeta[1/2, 23/24 + n], {n, Infinity, 2}]
2 Sqrt[n] + HurwitzZeta[1/2, 23/24] + (11 Sqrt[1/n])/24 - (73 (1/n)^(3/2))/2304 + O[1/n]^(5/2)
```

$$\exp\left(-\frac{3^{j/2}}{j\,\pi^{j}\,2^{j/2}}\sum_{k=1}^{n}\frac{1}{\left(k-\frac{1}{24}\right)^{j/2}}\right)\sim\exp\left(-\frac{3^{1/2}}{\pi\,2^{1/2}}*\left(2\sqrt{n}+\zeta\left(\frac{1}{2},\frac{23}{24}\right)\right)\right)=\exp\left(-\frac{\sqrt{6n}}{\pi}-\frac{3^{1/2}}{\pi\,2^{1/2}}\,\zeta\left(\frac{1}{2},\frac{23}{24}\right)\right)$$

Case $j=2$

```
Sum[1 / (k - 1 / 24), {k, 1, n}]
-PolyGamma[0, 23/24] + PolyGamma[0, 23/24 + n]

Series[-PolyGamma[0, 23/24] + PolyGamma[0, 23/24 + n], {n, Infinity, 2}]
(-Log[1/n] - PolyGamma[0, 23/24]) + 11/(24 n) - 73/(1152 n^2) + O[1/n]^3
```

```
Gamma'[23/24] / Gamma[23/24]
PolyGamma[0, 23/24]
```

$$\exp\left(-\frac{3^{j/2}}{j\,\pi^{j}\,2^{j/2}}\sum_{k=1}^{n}\frac{1}{\left(k-\frac{1}{24}\right)^{j/2}}\right)\sim\exp\left(-\frac{3}{4\,\pi^{2}}*\left(\log(n)-\frac{\Gamma'\left(\frac{23}{24}\right)}{\Gamma\left(\frac{23}{24}\right)}\right)\right)=n^{-\frac{3}{4\pi^{2}}}*\exp\left(\frac{3}{4\pi^{2}}\frac{\Gamma'\left(\frac{23}{24}\right)}{\Gamma\left(\frac{23}{24}\right)}\right)$$

For $j \geq 3$ we have

$$\lim_{n\to\infty} \zeta\left(\frac{j}{2}, n+\frac{23}{24}\right) = \lim_{n\to\infty} \sum_{m=0}^{\infty} \frac{1}{\left(m+n+\frac{23}{24}\right)^{j/2}} = 0$$

For example

```
Series[HurwitzZeta[3/2, 23/24] - HurwitzZeta[3/2, 23/24 + n], {n, Infinity, 2}]

HurwitzZeta[3/2, 23/24] - 2 Sqrt[1/n] + 11/24 (1/n)^(3/2) + O[1/n]^(5/2)

Series[HurwitzZeta[4/2, 23/24] - HurwitzZeta[4/2, 23/24 + n], {n, Infinity, 2}]

HurwitzZeta[2, 23/24] - 1/n + 11/(24 n^2) + O[1/n]^3
```

and

$$\exp\left(-\sum_{j=3}^{\infty} \frac{3^{j/2}}{j\,\pi^j\,2^{j/2}} \sum_{k=1}^{n} \frac{1}{\left(k-\frac{1}{24}\right)^{j/2}}\right) \sim \exp\left(-\sum_{j=3}^{\infty} \frac{3^{j/2}}{j\,\pi^j\,2^{j/2}} \zeta\left(\frac{j}{2}, \frac{23}{24}\right)\right)$$

The total contribution of Product 3 is

$$\prod_{k=1}^{n}\left(1-\frac{1}{\pi}\sqrt{\frac{3}{2\left(k-\frac{1}{24}\right)}}\right) = \exp\left(-\frac{\sqrt{6n}}{\pi} - \frac{3^{1/2}}{\pi\,2^{1/2}}\zeta\left(\frac{1}{2},\frac{23}{24}\right)\right) * \mathrm{n}^{-\frac{3}{4\pi^2}} * \exp\left(\frac{3}{4\pi^2}\frac{\Gamma'\left(\frac{23}{24}\right)}{\Gamma\left(\frac{23}{24}\right)}\right) * \exp\left(-\sum_{j=3}^{\infty}\frac{3^{j/2}}{j\,\pi^j\,2^{j/2}}\zeta\left(\frac{j}{2},\frac{23}{24}\right)\right)$$

Together, with a numerical computation of the constant C we obtain the final result.

$$C = \lim_{n\to\infty} \prod_{k=1}^{n} \frac{p(k)}{\dfrac{\exp\left(\pi\sqrt{2\,(k-1/24)/3}\right)}{4\sqrt{3}\,(k-1/24)} * \left(1-\dfrac{1}{\pi}\sqrt{\dfrac{3}{2\,(k-1/24)}}\right)}$$

Numerically, the iteration cycle:

```
Do[Print[Product[N[PartitionsP[k] / ((E^(Sqrt[2/3] Sqrt[k - 1/24] Pi) (1 - Sqrt[3/2]/(Sqrt[k - 1/24] Pi))) / (4 Sqrt[3] (k - 1/24))), 100], {k, 1, n}]], {n, 1000, 50000, 1000}]

0.9110167313322499526279306111304238284514104678400646686246815870467227055667326255557084446139708
0.9110167313322499518615475086149379685267490557322013111435245825494552190786073988089719162747336
0.9110167313322499518615474695947821117071922350578032007991370254612546092636530939397823737553445
0.9110167313322499518615474695946834546652973294020513859509590374274655534825802439457886695996440
0.9110167313322499518615474695946834527808697103798817775717560683536337667676186431544611473283393
0.9110167313322499518615474695946834527807386326233768294783029587250402583650782778776554987365222
0.9110167313322499518615474695946834527807386097880625360219364117655405275581853704129057310014409
0.9110167313322499518615474695946834527807386097800857682794707688125554065014280279070571202004550
0.9110167313322499518615474695946834527807386097800809348814365799116705448273990354798887526193944
0.9110167313322499518615474695946834527807386097800809302876977829131292181008142950419874194588244
```

After more iterations we get more decimal places

$C =$ A259314 $= 0.9110167313322499518615474695946834527807386097800809302813214902275914912 4\ldots$

Numerical verification (the asymptotics ratio):

```
A058694 = ParallelTable[Product[PartitionsP[k], {k, 1, n}], {n, 1, 10000}];
c = 0.911016731332249951861547469594683452780738609780080930281321490227591491240455575116502537070275392104475;
Show[
 ListPlot[
  Table[
   A058694[[n]]/
    (c*Gamma[23/24]/(n^(n+11/24+3/(4π^2)) 2^(2n) 3^(n/2) Sqrt[2π]) * Exp[π (2n/3)^(3/2) + n + (11π/(12 Sqrt[6]) - Sqrt[6]/π) * Sqrt[n] + Zeta[-1/2, 23/24] * Sqrt[2/3] π - Zeta[1/2, 23/24] * 1/π Sqrt[3/2] +
      3 Gamma'[23/24]/(4 π^2 Gamma[23/24]) - N[Sum[Zeta[j/2, 23/24] * (1/π Sqrt[3/2])^j/j, {j, 3, Infinity}], 20]]), {n, 1, Length[A058694]}]],
 Plot[1, {n, 1, Length[A058694]}, PlotStyle -> Red], PlotRange -> {0.99, 1}, AxesOrigin -> {0, 0.99}]
```



The family of factorial constants:
A062073 Fibonacci factorial constant
A218490 Lucas factorial constant
A253924 Padovan factorial constant
A256831 Pell factorial constant
A259314 partition factorial constant

References:

[1] OEIS - The On-Line Encyclopedia of Integer Sequences
[2] G. H. Hardy and S. Ramanujan, Asymptotic formulae in combinatory analysis, Proc. London Math. Soc., 17:75–115
[3] George E. Andrews, The Theory of Partitions, 1998
[4] M. Knopp, Modular Functions in Analytic Number Theory, 1970, p.90, Theorem 2
[5] Wikipedia, Integer partition
[6] Eric Weisstein's MathWorld, Partition Function P
[7] Eric Weisstein's MathWorld, Hurwitz Zeta Function

This article was published on the website http://web.telecom.cz/vaclav.kotesovec/math.htm, 26.6.2015
and in the OEIS, Jun 26 2015